



UN SÉQUENCEUR DANS VOTRE ORDINATEUR

Il y a deux mois, je vous avais proposé de réaliser un synthétiseur avec votre micro-ordinateur. Espérant que vous avez su en tirer le maximum de sons, je vous propose cette fois-ci de le compléter en lui ajoutant un séquenceur. Grâce à celui-ci, vous pourrez enregistrer une séquence sur trois voix pour votre accompagnement et chanter ou jouer d'un autre instrument lors de la restitution de votre enregistrement.

Ce programme ressemble de très près à une gestion de fichier. Dans un premier temps, on entre les sons. On peut ensuite écouter chaque voix séparément ou les trois en même temps. On peut enfin sauvegarder ou charger une séquence sur disquette ou cassette selon votre configuration. Mais avant d'attaquer le plat de résistance, nous allons commencer par apporter quelques modifications sur le programme AMSYNTH (Ams. Mag. n°16) afin de le rendre compatible avec la suite du programme. Tout d'abord le dimensionnement. A vrai dire, le nombre de variables n'est pas énorme. Seuls la période et la durée de la note vont être mémorisés. On va donc changer la ligne 60 :

```
60 DIM notes(13), touches(13), loca-
note(13), locatouche(13), per(300,3),
tempo(30,3)
```

Nous allons faire un ajout à la ligne 200 afin de mettre le séquenceur dans le menu :

```
200 LOCATE 8,20 : PRINT "T = ecran de
travail" : LOCATE 8,21 : PRINT "A =
ecran d'aide" : LOCATE 8,22 : PRINT "S
= sequenceur" : LOCATE 12,24 : PRINT
"Choisissez votre option"
```

Nous allons maintenant ajouter la ligne 235 afin d'aiguiller le programme vers le séquenceur.

```
235 IF inkey(60)=0 THEN 3000
```

Afin de faciliter l'entrée des notes, nous allons par la suite utiliser la fenêtre donnant la correspondance entre les notes et les touches du clavier. Il va donc falloir ajouter un "return" conditionnel afin de savoir si le programme doit continuer en synthétiseur ou retourner en séquenceur.

```
415 IF seq=0 THEN 420 ELSE RETURN
Le même problème va se poser à un autre
endroit du programme que nous verrons par
la suite. En attendant entrez la ligne 990 :
```

```
990 IF seq = 1 THEN RETURN : RE-
LEASE 7
```

Après relecture, je me suis aperçu qu'une fausse note s'était glissée à la ligne 2000. Après avoir accepté mes plates excuses entrez donc en 2000 no=676 à la place de no=376. Trois lignes sont encore à ajouter dans l'ancien programme. La première fixe la valeur de la période d'une note pendant un silence, la seconde ligne permet de retour-

ner vers le séquenceur s'il est utilisé et enfin la troisième sert à aller vers le séquenceur à l'aide de la touche copy, lorsque l'on est en synthétiseur :

```
2186 IF inkey (47) = 0 THEN no = 0 : gosub 990
```

```
2187 IF seq = 1 THEN RETURN
```

```
1755 IF inkey (9) = 0 THEN 3000
```

Voici donc terminées les modifications et ajouts à faire sur le programme d'origine. Nous allons donc enfin passer aux choses sérieuses avec le début du séquenceur. Celui-ci commence à 3000 afin de pouvoir être ajouté directement derrière AMSYNTH. La totalité du programme fera alors 19 KO ce qui laisse encore de la marge pour d'éventuels ajouts. Tout d'abord, afin d'éviter tout problème, repassons la vitesse de balayage du clavier à une vitesse normale. C'est ce qui est fait à la ligne 3030.

Vient ensuite le traditionnel menu où vous sont offerts l'enregistrement, l'audition, le chargement et enfin la sauvegarde d'une séquence sans oublier le retour vers le synthétiseur. Je passe rapidement sur ces lignes qui ne présentent aucune difficultés.

Nous voici maintenant dans la partie du programme permettant d'enregistrer une séquence. Au début, je définis des fenêtres qui vont me permettre de séparer l'écran pour les instructions, le clavier et les trois canaux. Je vais ensuite chercher le clavier dans le programme de base à la ligne 320.

A ce moment là, vient un choix très important avec le choix du tempo. Il suffit d'entrer la valeur de la note, valeur annotée sur la majorité des partitions. Viennent quelques instructions qui vous seront très utiles pour la bonne marche de l'enregistrement. En 3320 le programme demande de choisir le canal sur lequel on veut travailler. L'ordre est absolument indifférent et rien n'empêche de commencer par le canal 3 pour aller ensuite au 1 et finir par le 2. En 3360 j'indique à l'écran quel est le canal sur lequel on travaille ce qui est surtout utile en cas de changement que vous sêrte sûr que le micro a bien suivi les ordres. Faisant tourner trois compteurs parallèlement pour les trois canaux, j'indique en 3370-80-90 quel est le compteur qui doit se mettre en route. Vient ensuite la saisie de la note sur le clavier puis la recher-

che de sa période dans le synthétiseur. En effet, plutôt que de stocker une grandeur alphanumérique et d'en faire la conversion avant chaque audition, je trouve préférable de traduire tout de suite la valeur en sa réelle valeur de période. Arrivé ici, le travail de la période est presque terminé. Néanmoins, pour plus de clarté, il est préférable de lire à l'écran la valeur de la note que l'on vient d'entrer. Je vais donc chercher dans un tableau situé en 4410 la correspondance entre la période et le nom de la note. Cette fois-ci, tout le travail de la période est fait et il ne me reste plus qu'à saisir la durée de la note. La saisie se fait de la même manière avec les touches du bas du clavier :

```
z = ronde
```

```
x = blanche
```

```
c = noire
```

```
v = croche
```

```
b = double croche
```

```
m = triple croche
```

```
n = triolet
```

```
shift z = ronde pointée
```

```
shift x = blanche pointée
```

```
shift c = noire pointée
```

```
shift v = croche pointée
```

Une fois la touche enregistrée, le programme va en 4270 pour calculer la durée de la note selon la valeur de la note. Ici même démarque pour la période. Je transforme tout de suite la durée et mémorise cette valeur. Afin de simplifier l'emploi de ce séquenceur, je ne peux que trop vous conseiller d'utiliser des petites étiquettes auto-collantes pour les touches. Cela vous évitera des instants de réflexions inutiles et douloureux.

Avant d'en finir avec la saisie des notes un mot sur les silences. Ceux-ci sont obtenus avec la barre espace suivi de leur valeur temporelle. Pour changer de canal, frappez la touche ENTER du pavé numérique puis le numéro du nouveau canal. Lorsque la saisie est finie, frappez la grosse touche ENTER pour revenir au menu.

Votre œuvre est achevée et vous voulez l'écouter, allez donc en 3570 et choisissez à nouveau la carte. Dans le cas d'une écoute monodique, je commence par établir des correspondances entre le numéro du canal et son numéro réel. Rappelez-vous que les trois canaux sont 1, 2, 4 et que canal 3 = canal 1 + canal 2. Donc pas de confusion ici car je veux vraiment trois canaux indépendants. A ce stade, soit je choisis l'écoute de la totalité du morceau sur la voix choisie. Dans ce cas, pas de problème. Une petite boucle et le tour est joué ; soit je choisis l'option note à note. Cela se complique mais me permet de suivre plus attentivement ce que je viens de rentrer pour éventuellement le modifier. Dans le cas d'une modification, tapez M, puis la nouvelle note de la même manière que dans la saisie, puis S pour passer à la note suivante. Si une note a été entrée deux fois à la place d'une, tapez E pour effacer l'intruse. Lorsque tout vous semble parfait tapez R pour revenir au menu et éventuellement écouter les trois voix ensemble. Enfin, pour terminer nous arrivons aux

TRUCS ET BICOUILLES

modules de sauvegarde et de chargement. Dans les deux cas, rien de très difficile. Juste un petit inventaire des variables que je dois absolument sauvegarder et le tour est joué. Bien que je ne l'ai pas fait ici, vous pourriez étendre la sauvegarde aux sons qui pourraient être spécifiques à une musique donnée.

En effet n'oubliez pas que les sons joués sont les sons du programme de synthèse de son. PS : Pour terminer je voudrais justement revenir sur ce programme de synthèse. Il semble que plusieurs personnes aient eu des problèmes. Alors plutôt que de téléphoner à la rédaction d'Amstrad Magazine où je ne

suis jamais, écrivez-moi au journal en m'exposant vos problèmes et vos éventuelles suggestions pour de futurs articles et programmes. A vos claviers donc pour la saisie de cette deuxième partie qui, je peux vous le jurer, tout comme la première fonctionne.

Dominique Gourdière

```
3000 ***** sequenceur *****
3010 seq=1
3020 seq$="S E Q U E N C E U R"
3030 SPEED KEY 20,3
3040 n=1:a=1:b=1:d=1
3050 l=LEN(seq$):ll=(40-1)/2
3060 MODE 1:LOCATE 11,5:PRINT inv$+seq$+inv$
3070 LOCATE 5,10:PRINT" 1 - enregistrement d'une sequence"
3080 LOCATE 5,12:PRINT" 2 - audition d'une sequence"
3090 LOCATE 5,14:PRINT" 3 - chargement d'une sequence"
3100 LOCATE 5,16:PRINT" 4 - sauvegarde d'une sequence"
3110 LOCATE 5,18:PRINT" 5 - retour au synthetiseur"
3120 LOCATE 5,24:PRINT" Entrez votre choix"
3130 a$=INKEY$:IF a$="" THEN 3130
3140 IF a$="1" THEN 3200
3150 IF a$="2" THEN 3560
3160 IF a$="3" THEN 4090
3170 IF a$="4" THEN 4180"
3180 IF a$="5" THEN seq=0:GOTO 250
3190 GOTO 3130
3200 ***** enregistrement d'une sequence *****
3210 MODE 2
3220 WINDOW #1,1,80,1,4:WINDOW #7,1,80,21,25:WINDOW #2,1,80,5,20
3230 WINDOW #4,1,27,6,20:WINDOW #5,28,54,6,20:WINDOW #6,55,80,6,20
3240 CLS:GOSUB 320
3250 INPUT #1,"Entrez la valeur de la noire ":valnoire:CLS #1
3260 LOCATE #1,3,1:PRINT#1,"Entrez la note puis sa valeur"
3270 LOCATE #1,3,2:PRINT#1,"Pour les silences,tapez barre espace puis sa valeur"
3280 LOCATE #1,3,3:PRINT#1," ENTER pour sortir"
3290 :LOCATE #2,1,1:PRINT#2,STRING$(80,"_"):LOCATE #2,1,15:PRINT#2,STRING$(80,
" ")
3300 FOR i=1 TO 14:LOCATE #5,1,i:PRINT#5,"*":LOCATE #5,26,i:PRINT#5,"*":NEXT
3310 FOR i=1 TO 3:LOCATE #3+i,8,2:PRINT#3+i,"CANAL ";i:NEXT
3320 a$=INKEY$:IF a$="" THEN 3320
3330 IF INKEY(13)=0 THEN c=1
3340 IF INKEY(14)=0 THEN c=2
3350 IF INKEY(5)=0 THEN c=3
3360 w=c+3:LOCATE #w,2,3:PRINT#w,"canal en travail"
3370 IF c=1 THEN n=a
3380 IF c=2 THEN n=b
3390 IF c=3 THEN n=d
3400 a$=INKEY$:IF a$="" THEN 3400
3410 LOCATE #w,5,5:PRINT#w,STRING$(8,CHR$(32))
3420 GOSUB 1800
3430 IF INKEY(18)=0 THEN 3060
3440 IF INKEY(6)=0 THEN LOCATE #w,2,3:PRINT#w,STRING$(20,CHR$(32)):GOTO 3320
3450 per(n,c)=no
3460 GOSUB 4410
3470 a$=INKEY$:IF a$="" THEN 3470
3480 GOSUB 4270
3490 SOUND c,per(n,c),tempo(n,c)
3500 IF modif =1 THEN RETURN
3510 LOCATE #w,3,13:PRINT#w,"RESTE ":300-n;" NOTES"
3520 IF c=1 THEN a=n
```



```

3530 IF c=2 THEN b=n
3540 IF c=3 THEN d=n
3550 n=n+1:GOTO 3400
3560 '***** audition d'une sequence *****
3570 MODE 1
3580 LOCATE 5,10:PRINT"1 - Audition du canal 1"
3590 LOCATE 5,12:PRINT"2 - Audition du canal 2"
3600 LOCATE 5,14:PRINT"3 - Audition du canal 3"
3610 LOCATE 5,16:PRINT"4 - Audition des 3 canaux"
3620 LOCATE 5,18:PRINT"5 - Retour au menu"
3630 LOCATE 5,20:PRINT"Entrez votre choix (1,2,3,4,5)"
3640 a#=INKEY#:IF a#="" THEN 3640
3650 q=VAL(a#):canal=q:IF canal=3 THEN canal=4
3660 IF q=4 THEN 3920
3670 IF q=5 THEN 3060
3680 CLS:LOCATE 5,10:PRINT"1 - Audition normale"
3690 LOCATE 5,12:PRINT"2 - Audition note / note"
3700 a#=INKEY#:IF a#="" THEN 3700
3710 r=VAL(a#):ON r GOTO 3720,3750
3720 FOR i=1 TO a:SOUND canal,per(i,q),tempo(i,q),v(q),q,q,b(q)
3730 NEXT
3740 GOTO 3560
3750 MODE 2
3760 LOCATE 3,3:PRINT"S = note suivante - M = modification - R = retour -
E = effacement"
3770 WINDOW #7,1,80,21,25:GOSUB 320
3780 IF q=1 THEN mx=a:IF q=2 THEN mx=b:IF q=3 THEN mx=d
3790 i=1
3800 SOUND canal,per(i,q),tempo(i,q),v(q),q,q,b(q)
3810 a#=INKEY#:IF a#="" THEN 3810
3820 IF INKEY(60)=0 THEN i=i+1:IF i>mx THEN 3560 ELSE GOTO 3800
3830 IF INKEY(50)=0 THEN 3560
3840 IF INKEY(38)=0 THEN 3870
3850 IF INKEY(58)=0 THEN FOR j=i TO mx:per(j,q)=per(j+1,q):tempo(j,q)=tempo(j+1,
q):NEXT:mx=mx-1:GOTO 3800
3860 FOR j=mx TO i:per(j+1,q)=per(j,q):tempo(j+1,q)=tempo(j,q):NEXT
3870 LOCATE 5,7:PRINT"entrer la nouvelle note ";
3880 modif=1: c=q:n=i:LOCATE 32,5:GOSUB 3400
3890 LOCATE 5,7:PRINT STRING$(40,CHR$(32))
3900 i=i+1:
3910 GOTO 3810
3920 RELEASE 7:x=0:y=0:z=0
3930 ON SQ(1) GOSUB 3970
3940 ON SQ(2) GOSUB 4010
3950 ON SQ(4) GOSUB 4050
3960 GOTO 3560
3970 x=x+1:IF x>a THEN RETURN
3980 SOUND 1,per(x,1),tempo(x,1),v(1),1,1,b(1)
3990 ON SQ(1) GOSUB 3970
4000 RETURN
4010 y=y+1:IF y>b THEN RETURN
4020 SOUND 2,per(y,2),tempo(y,2),v(2),2,2,b(2)
4030 ON SQ(2) GOSUB 4010
4040 RETURN
4050 z=z+1:IF z>d THEN RETURN
4060 SOUND 4,per(z,3),tempo(z,3),v(3),3,3,b(3)
4070 ON SQ(4) GOSUB 4050
4080 RETURN
4090 '***** changement d'une sequence *****
4100 CLS:LOCATE 5,5:INPUT"Nom du morceau a charger ":n$
4110 OPENIN n$

```



```

4120 INPUT #9,a,b,d
4130 FOR i=1 TO a:INPUT #9,per(i,1),tempo(i,1):NEXT
4140 FOR i=1 TO b:INPUT #9,per(i,2),tempo(i,2):NEXT
4150 FOR i=1 TO d:INPUT #9,per(i,3),tempo(i,3):NEXT
4160 CLOSEIN
4170 GOTO 3060
4180 ?***** sauvegarde d'une sequence *****
4190 CLS:LOCATE 5,5:INPUT "Nom du morceau a sauvegarder ";n$
4200 OPENOUT n$
4210 PRINT#9,a,b,d
4220 FOR i=1 TO a:PRINT#9,per(i,1),tempo(i,1):NEXT
4230 FOR i=1 TO b:PRINT#9,per(i,2),tempo(i,2):NEXT
4240 FOR i=1 TO d:PRINT#9,per(i,3),tempo(i,3):NEXT
4250 CLOSEOUT
4260 GOTO 3060
4270 ? ***** definition de la valeur de la note *****
4280 noire = 6000/valnoire
4290 IF INKEY(71)=0 THEN tempo(n,c)=noire *4:PRINT#w,"R"
4300 IF INKEY(63)=0 THEN tempo(n,c)=noire *2:PRINT#w,"B"
4310 IF INKEY(62)=0 THEN tempo(n,c)=noire:PRINT#w,"N"
4320 IF INKEY(55)=0 THEN tempo(n,c)=noire /2:PRINT#w,"C"
4330 IF INKEY(54)=0 THEN tempo(n,c)=noire /4:PRINT#w,"DC"
4340 IF INKEY(46)=0 THEN tempo(n,c)=noire /8:PRINT#w,"TC"
4350 IF INKEY(71)=32 THEN tempo(n,c)=noire *6:PRINT#w,"RP"
4360 IF INKEY(63)=32 THEN tempo(n,c)=noire *3:PRINT#w,"BP"
4370 IF INKEY(62)=32 THEN tempo(n,c)=noire *1.5:PRINT#w,"NP"
4380 IF INKEY(55)=32 THEN tempo(n,c)=noire *0.75:PRINT#w,"CP"
4390 IF INKEY(38)=0 THEN tempo(n,c)=noire /3:PRINT#w,"TR"
4400 RETURN
4410 ? ***** corellation periode / nom de la note *****
4420 IF per(n,c)=478 OR per(n,c)=239 OR per(n,c)=956 OR per(n,c)=119 THEN LOCATE
#w,5.5:PRINT#w,"D0,";
4430 IF per(n,c)=451 OR per(n,c)=225 OR per(n,c)=902 THEN LOCATE #w,5.5:PRINT#w,
"D0#,";
4440 IF per(n,c)=426 OR per(n,c)=213 OR per(n,c)=851 THEN LOCATE #w,5.5:PRINT#w,
"RE,";
4450 IF per(n,c)=402 OR per(n,c)=201 OR per(n,c)=804 THEN LOCATE #w,5.5:PRINT#w,
"RE#,";
4460 IF per(n,c)=379 OR per(n,c)=190 OR per(n,c)=758 THEN LOCATE #w,5.5:PRINT#w,
"MI,";
4470 IF per(n,c)=358 OR per(n,c)=179 OR per(n,c)=716 THEN LOCATE #w,5.5:PRINT#w,
"FA,";
4480 IF per(n,c)=338 OR per(n,c)=169 OR per(n,c)=676 THEN LOCATE #w,5.5:PRINT#w,
"FA#,";
4490 IF per(n,c)=319 OR per(n,c)=159 OR per(n,c)=638 THEN LOCATE #w,5.5:PRINT#w,
"SOL,";
4500 IF per(n,c)=301 OR per(n,c)=150 OR per(n,c)=602 THEN LOCATE #w,5.5:PRINT#w,
"SOL#,";
4510 IF per(n,c)=284 OR per(n,c)=142 OR per(n,c)=568 THEN LOCATE #w,5.5:PRINT#w,
"LA,";
4520 IF per(n,c)=268 OR per(n,c)=134 OR per(n,c)=526 THEN LOCATE #w,5.5:PRINT#w,
"LA#,";
4530 IF per(n,c)=253 OR per(n,c)=127 OR per(n,c)=506 THEN LOCATE #w,5.5:PRINT#w,
"SI,";
4540 IF per(n,c)=0 THEN LOCATE #w,5.5:PRINT#w,"SIL,";
4550 RETURN
4560 x=1
4570 SOUND 1,per(x,1),tempo(x,1)
4580 a$=INKEY$:IF a$="" THEN 4580
4590 IF INKEY(60)=0 THEN x=x+1
4600 GOTO 4570

```